Newsgroup: it.comp.lang.visual-basic
Titolo: Creare un foglio Excel tramite DAO

Salve a tutti,
qualcuno mi saprebbe gentilmente fare un esempio su come creare un file di
Excel da Vb usando DAO ?
Grazie in anticipo per l'eventuale risposta.
--
Ciao
Roberto
Email:
robert...@libero.it

 Grazie in anticipo per l'eventuale risposta.
devo avere un esempio che prevede una dbgrid e excel installato.
se vuoi lo cerco.

se vuoi lo cerco.
Io ho trovato questo:
Nei riferimenti devi aggiungere Microsoft Excel 9.0 object library (o
8.0 ...?!?!?)
ti cosiglio di farti il tuo normale ciclo di lettura del recordset e
poi invece che riempire una form, un report o che altro riempi il
foglio di excel .. vedi se riescio a capirci qualcosa dal codice qui
sotto e aiutati con la guida in linea !!!
------------------
Dim prove As excel.Application
 Set prove = CreateObject("excel.application")
  prove.Visible = True
  prove.Workbooks.Add
  prove.Cells(2, 2)="ciao"
  prove.Workbooks(1).Worksheets("foglio1").Range(strcol) = "=SUM(B" &
i + 6 & ":Z" & i + 6 & ")"
  prove.Workbooks(1).Worksheets("foglio1").Range("A6:C23").ColumnWidth
= 2
  prove.Workbooks(1).Worksheets("foglio1").Range("A6:C23").Font.Size =
7
  prove.Workbooks(1).Worksheets("foglio1").Range
("A6:C23").Borders.Color = QBColor(8)
  prove.Workbooks(1).Worksheets("foglio1").Range
("A6:C23").Borders.LineStyle = xlDot
  prove.Workbooks(1).Worksheets("foglio1").Range
("A6:C23").Borders.Color = RGB(0, 0, 0)
  prove.Workbooks(1).Worksheets("foglio1").Range
("A6:C23").Borders.LineStyle = xlContinuous
  prove.Workbooks(1).Worksheets("foglio1").Range
("A6:C23").).Borders.Color = RGB(0, 0, 0)
    prove.Workbooks(1).Sheets.Add
after:=prove.Workbooks(1).ActiveSheet
    prove.Workbooks(1).Sheets(index).Name ="PIPPO"
  prove.DisplayAlerts = False
  prove.Workbooks(1).SaveAs (Form3.Text1.Text)
  prove.Workbooks(1).Close
  prove.Quit
------------------
Ciaps
Stracca
http://www.stracca.net
---------------------------------------------
    Le risposte alle tue domande le trovi
    sul Sito Comune di questo NewsGroup:
 
http://www.murialdo.it/it_lang_vb
       ... e nell'help in linea !
---------------------------------------------

Vi ringrazio per il vostro interessamento.
A presto....


 qualcuno mi saprebbe gentilmente fare un esempio su come creare un file di
Excel da Vb usando DAO ?
' TBL2WKS
'
Option Explicit
Public Sub Main( _
)
  Const c_strFile  As String = "Northwind"
  Const c_strDbs   As String = "C:\Programmi" _
                             & "\Microsoft Office" _
                             & "\Office" _
                             & "\Samples" _
                             & "\" & c_strFile & ".mdb"
  Const c_strTable As String = "Clienti"
  Dim dbs     As DAO.Database
  Dim rst     As DAO.Recordset
  Dim xlApp   As Excel.Application
  Dim xlWbk   As Excel.Workbook
  Dim xlWks   As Excel.Worksheet
  Dim strWbk  As String
  Dim intFld  As Integer
  Dim lngRow  As Long
  Set dbs = DAO.DBEngine.OpenDatabase(c_strDbs)
  Set rst = dbs.OpenRecordset(c_strTable, dbOpenTable)
  Set xlApp = New Excel.Application
  Set xlWbk = xlApp.Workbooks.Add
  Set xlWks = xlWbk.ActiveSheet
  xlWks.Name = c_strTable
  With rst
    With .Fields
      lngRow = 1
      For intFld = 0 To .Count - 1
        xlWks.Cells(lngRow, intFld + 1).Value _
          = .Item(intFld).Name
      Next
    End With
    If .BOF And .EOF Then
    Else
      Do While Not .EOF
        With .Fields
          lngRow = lngRow + 1
          For intFld = 0 To .Count - 1
            xlWks.Cells(lngRow, intFld + 1).Value _
              = .Item(intFld).Value
          Next
        End With
        .MoveNext
      Loop
    End If
    .Close
  End With: Set rst = Nothing
  dbs.Close
  Set dbs = Nothing
  Set xlWks = Nothing
  strWbk _
    = App.Path _
    & IIf(Right$(App.Path, 1) = "\", "", "\") _
    & c_strFile & ".xls"
  If Len(Dir$(strWbk)) Then Kill strWbk
  xlWbk.Close True, strWbk
  Set xlWbk = Nothing
  xlApp.Quit
  Set xlApp = Nothing
End Sub
--
Ciao.
Maurizio
--------
?SPQR(C)
 X
--------